
On Daisyworlds
A personal account on the reconstruction

of a model to test the Gaia hypothesis
by

Andreas Fischlin1

Zürich, March 1990

Abstract
Lovelock (1988) states the so-called Gaia hypothesis, which postulates that
the surface of the globe forms an ecosystem which can be considered to be a
living super-organism.  To demonstrate that this argument is neither teleologi-
cal nor mystical, he has constructed a simple mathematical model.  It simulates
the competition of two daisy species, each with a different colour and hence a
different albedo.  The global albedo varies in function of the current species
composition and thus the daisies become capable of regulating homoeostati-
cally the global temperature of this Daisyworld.  This text describes in more
or less chronological order an attempt to reconstruct the mathematical equa-
tions of a Daisyworld model, since in his book Lovelock gives no precise
description of such a model, nor a proper reference to a more detailed article.
The purpose of this exercise was to test the robustness of Lovelock's
conclusions, by reconstructing a Daisyworld model independently and from
scratch, solely in accordance with the basic assumptions Lovelock has
sketched in his book.  After an initial, but minor parameter fitting, the
behaviour of the newly derived model matches the results Lovelock describes.
Thus, the homeostatic property of a Daisyworld as postulated by the Gaia
hypothesis appears to be fairly robust and independent of a particular
mathematical formulation, for instance that chosen by W ATSON and
LOVELOCK (1983).

Introduction

In his new book LOVELOCK (1988) postulates once more the so-called Gaia hypothesis
(LOVELOCK, 1972; LOVELOCK & MARGULIS, 1974; MARGULIS & LOVELOCK, 1974).
It amounts to the idea that the biosphere of the globe together with the biota form a
homeostatic system which controls its environment and which can be considered to be a
living super-organism.  His idea got from enthusiastic appraisal and complete
condemnation (e.g. POSTGATE, 1988) an impressively vast range of responses, at least a
lot of attention (SCHNEIDER, 1989; LOVELOCK, 1990).  However, it seems to me to be
obvious that the Gaia hypothesis as formulated by LOVELOCK (1988), actually
encompasses several, distinct issues.  Each issue can be investigated independently from
the other, thus allowing to give Lovelock merit where he deserves it and to challenge him
where he might really fail to make a convincing and scientifically sound argument.  The
three issues are:  The biosphere forms a super-ecosystem on a global scale; the biosphere
is alive; the biosphere has the capability to regulate homoeostatically its environment to its
own profit.  

Concerning the first issue, Lovelock adopts a terminology, which requires some
clarification.  Unfortunately he contributes once more to the already big enough confusion
in regard to the term biosphere:  He uses it as the topic entity (SCHWERDTFEGER, 1975),
that is the space which life occupies; but he does not clearly separate it from the biota, i.e.
all living organisms together, which form the biotic entity (SCHWERDTFEGER, 1975).
According to the meaning Lovelock attaches to the term Gaia, it is the ecological entity, i.e.
the global system formed by the topic and biotic entities together.  On a smaller spatial
scale the term ecosystem is well established and almost unanimously considered to be the
ecological entity, i.e. it encompasses first the habitat as the topic entity and second the
biocoenosis as the biotic entity.  Ecologists prefer to adopt a similar terminology on the
global scale.  Accordingly, in contrast to Lovelock, I will call Gaia the ecosphere, it
encompasses a topic and a biotic entity:  the topic entity is called toposhere and the biotic
biosphere.  GILLARD (1969) and SCHWERDTFEGER (1975) have proposed this scheme
and it seems to offer many advantages.  For instance, analogies between synecological
theory and global ecology are much easier to postulate, than this would be possible
otherwise, and arguments made on a global scale are easier to understand clearly.  E.g., if
Lovelock refers in an argument to the oceanic biosphere, one often does not know what he
really means; is it now the whole surface layer of the oceans, i.e. the photic zone, which is
a distinctly structured water layer of about 100 m depth, or the biomass composed of algae
and zooplancton living in this layer?

Of course, more importantly than terminology is the meaning of the term Gaia.  Note, the
same basic functions which are attributed to an ecosystem, such as a lake or a forest, can
be found again in the ecosphere:  The structuring into several trophic levels like plants,
consumers, and destruents; the essential functions like primary production and the
subsequent flow of energy through the trophic structure; an, at least to some degree,
autonomous metabolism such as nutrient cycling, e.g. the global carbon cycle2.  Hence,
nothing seems to be wrong with the statement: "the biosphere forms a super-ecosystem on
a global scale".  By the way, Lovelock uses a nice metaphor to describe these
relationships:  Gaia resembles a snail, which consists of a shell (abiotic, topic entity like
the lithosphere, atmosphere, and hydrosphere) and a body (biota or biosphere).  

                                                
2On the contrary, the ecosphere is even better bounded than this is the case for many classical
ecosystems; the latter have usually fuzzy borders only, the so-called ecotones and as a consequence it is
often difficult to define precisely where the ecosystems starts and ends.  E.g. consider the gradual
differences between forests with a closed canopy, open forests (crowns of trees don't touch each other any
more), savannah, bush- and grasslands.

Lovelock claims that the global ecosystem Gaia is alive.  He fails to give a clear definition
of alive.  Biology is the science of life, and it might provide us with better clues about its
meaning:  For biologists to be alive is a property, which is typically attributed to particular
beings such as animals or trees only, not to just any conceivable system.  Applied to such
easily recognizable entities as organisms, to be alive represents a rather well defined
property which requires that at least the following three characteristics hold:  A living
system ought to be capable of self-replication, a metabolism, and a modifiable inheritance
(@KAPLAN, 19xx).  Some authors (CZIHAK et al., 1976) assert the additional criterion of
a cellular structure.  In any case, alive can be defined in a way which has some specific
meaning in biology and which matches the intuitive, every-day understanding, and it
appears to be a term, which has to be and has been defined almost axiomatically3.
However, Lovelock seems to equate the state of a system to be alive with its homeostasis.
Now, the latter might be a property of some living systems, yet it is neither an obvious nor
an unanimously understood property of living systems.  Thus Lovelock's claim that the
ecosphere Gaia is alive, seems to be based on shaky grounds, i.e. it is merely based on a
personal redefinition of the term, which ignores the usual meaning.

Yet, less questionable is the statement, that the ecosphere is actually capable to regulate
homoeostatically its environment to its own profit, i.e. the homeostasis of the ecosphere.
This statement appears to be the most interesting and the least trivial part of the Gaia
hypothesis.  Lovelock himself follows a viable procedure to test and corroborate it:  He
claims to have constructed a simple mathematical model, called Daisyworld, built from
simple, testable elements, and that the hypothesized homeostasis only emerges from the
overall system behaviour.  According to Lovelock this demonstrates that the Gaia
hypothesis is neither teleological nor mystical.  As described in the book, Lovelock's
model is funded on the following basic assumptions:

Daisyworld consists of a sole continent with a surface which can be populated by several
daisy species.  Each daisy species has a different colour and hence a different albedo4.
Bare ground is described to have an albedo in-between the lightest and darkest daisy
species.  Depending on the current species composition of the daisies, the global albedo
differs and thus the daisies can influence the global earth temperature5.  The solar
luminosity increases as the sun ages, and leads to a relatively rapid change in the global
temperature regimes.

Lovelock presents several graphs (Fig. 2.1, p.38; Fig. 3.2, p.47; Fig. 3.3, p.49; Fig. 3.4,
p.53; and Fig. 3.5, p.55) showing that the daisies are capable to regulate the earth's
temperature in a relatively narrow range of 20-30 °C during a long period of time.
Without this mechanism of temperature regulation, the period with a viable temperature
range allowing the daisies to live would be much shorter.  In the course of billions of
years, Daisyworld is therefore capable of a homeostatic temperature regulation by
adapting its species composition such that the community of daisies can live in a more
favourable temperature regime.  However, it is not obvious that homeostasis follows
necessarily from just the assumptions stated above.  Further assumptions, in particular on
the specific mathematical formulation of the here described relationships ought be needed,
in order to conclusively derive or associate the homeostatic property with the basic
assumptions of the Gaia hypothesis.  

                                                
3For an excellent essay on the definition of life read MONOD (1970).
4The albedo α of a surface is defined as the reflected fraction of an incident radiation s expressed as the
ratio between the reflected and the incoming radiation measured in Wm-2.  
5Between temperature T in °K of a surface, its albedo α , and the radiation incidence hold certain physical
relationships, i.e. at equilibrium between absorbed and emitted radiation holds Kirchhoff's law:  
S(1 - α) = ε τ s T4, where for our world S is the solar constant, ε  is the infrared emissivity of the
atmosphere, τ is the transmissivity of the atmosphere, and s is the Stefan-Boltzmann constant.

Unfortunately, Lovelock does not describe the mathematics behind the daisy model; he
only mentions a paper he has written, but fails to give a proper reference and it took me
several weeks before I finally succeeded to locate that work (WATSON & LOVELOCK,
1983).  In the meantime I was so frustrated, I decided to try it on my own.  Moreover, I
felt even more challenged to construct my own daisy model, because of the following
reasons:  First, Lovelock light-heartedly expresses his jubilation about the robustness of
his daisy model, a property he claims to be in sharp contrast with all other results of
modern ecological theory (see e.g. GARDNER & ASHBY, 1970; MAY, 1972, 1974; PIMM,
1984).  Second, Lovelock tries to impress the reader with the claim that he is the first
author considering abiotic factors to play an important role in ecosystems.  He mentions a
single exception, i.e. E.P. Odum (ODUM, 1970), but at least to me, it appears rather
difficult to actually ignore the vast ecological literature which addresses this issue properly
(e.g. WEBSTER et al., 1974; the many references cited in ODUM, 1970; KREBS, 1972;
SCHWERTDFEGER, 1975).  

Finally, the exercise appeared to be worth-wile, in particular because I wanted also to find
out whether the robustness of Lovelock's daisy model was due to the assumptions of the
Gaia hypothesis or rather due to the particular structure of the equations WATSON and
LOVELOCK (1983) employed.  Lovelock's basic assumptions offer still much freedom in
the mathematical formulation of the systems particular properties and the essential
relationships between the daisy populations and the temperature of the planet Gaia6.  The
latter question seems to me to be of much greater importance than Lovelock admits, who
favours the idea that all what matters is the basic assumptions of the Gaia hypothesis as
described above.  He claims, that they are fully and sufficiently responsible for all this nice
robustness of the ecosphere Gaia, in particular if the species diversity becomes bigger than
this is the case on Daisyworld.  The latter is certainly true on our globe, but, Lovelock's
expectation is another, among ecologists highly debated topic and therefore simply
another questionable assertion7, I prefer not to discuss here.  

Thus the main question addressed in this paper is whether the homeostasis postulated by
the Gaia hypothesis is really as robust as Lovelock claims.  In a first step to test this I
attempted to falsify it, by constructing a daisyworld model which is based on the same
basic assumptions but otherwise derived independently.  On the other hand, the more
daisyworld models, which are mathematically formulated differently, exhibit the postulated
homeostatic behaviour, the more likely it becomes that the Gaia hypothesis is correct.  If
homeostasis should emerge from this first model, Lovelock's Gaia hypothesis could at
least be interpreted as being so robust as not to depend on the idiosyncrasies and
specialities of the particular mathematical formulation WATSON and LOVELOCK (1983)
employed.

Model derivation

It seemed to me to be appropriate to choose a model as simple as possible8 to follow a
top-down approach with step-wise refinement along the way.  I started with the

                                                
6For instance, as it is well known in systems theory, the existence of a negative feed-back loop is
necessary but not sufficient to warrant the stability of a system if it contains additional feed-back loops or
if it is non-linear; additional system properties such as specific parameter values or the exact mathematical
structure are required to determine the stability of a steady-state.  Note, the existence of a non-trivial,
stable steady-state is a necessary prerequisit of homoeostasis.  
7The issue of diversity and stability has been debated strongly within ecology; see e.g. @PIMENTEL,
MAY, MAYNARD-SMITH, ELLENBERG, HOLLING.
8 You see that I sympathize with Occam's razor (For an informative as well as amusing explanation of
Occam's razor I recommend HUTCHINSON, 1978)

a s s u m pti o n t h at p o p ul ati o n d e nsiti es o u g ht  t o  b e  t h e  s a m e  f or  all s p e ci es a n d  c o nst a nt
t hr o u g h o ut ti m e.  Fr o m t hi s f oll o w s t h at d ai si e s c o v er a c ert ai n p orti o n of t h e gl o b e's
c o nti n e nt o n d ais y w orl d, w hi c h ar e gi v e n s ol el y  b y  t h e  c urr e nt p o p ul ati o n si z es.    B ar e
gr o u n d is p o p ul at e d as t h e p o p ul ati o n gr o w, till t h er e is n o m or e s p a c e a v ail a bl e a n y m or e.
T his sit u ati o n m at c h es e x a ctl y t h e n oti o n of t h e c arr yi n g c a p a cit y as us e d i n t h e e q u ati o n
of l o gisti c gr o wt h.  N or m all y l o gisti c gr o wt h  is f or a si n gl e s p e ci es o nl y; b ut it w as e as y
t o d eri v e a v ari a nt all o wi n g f or c o m p etiti o n b et w e e n s p e ci es a n d still f oll o wi n g a  gr o wt h
pr o c ess wit h s at ur ati o n o n c e t h e c arr yi n g c a p a cit y is r e a c h e d:

x i=r i

K- x j∑
j = 1

n

K
x i ∀ i s p e ci es ( 1)

w h er e x i d e n ot es t h e p o p ul ati o n si z e of t h e i-t h s p e ci es, K t h e c arr yi n g c a p a cit y gi v e n b y
t h e si z e of t h e c o nti n e nt's s urf a c e a n d ri t h e s p e ci es s p e cifi c p er c a pit a gr o wt h r at e of t h e i-
t h d ais y s p e ci es.  

H o w e v er, i n c o ntr ast t o t h e cl assi c al l o gisti c e q u ati o n, t h e gr o wt h r at e r is n ot a  c o nst a nt
m o d el p ar a m et er, b ut a n a u xili ar y v ari a bl e:  I n a c c or d a n c e wit h t h e  G ai a  h y p ot h e si s  I
ass u m e d t h at s p e ci es d o gr o w i n f u n cti o n of t h e t e m p er at ur e r e g i m e, i. e. r is a f u n cti o n of
t h e c urr e nt t e m p er at ur e T.  I n r e alit y all p oi kil ot h er mi c s p e ci es s h o w a gr o wt h r at e ri w hi c h
d e p e n d s o n t e m p er at ur e i n a s p e ci es s p e cifi c m a n n er, si mil ar t o t h e f u n cti o n s h o w n i n
Fi g. 1.  B el o w T mi n/i , t h e mi ni m u m d e v el o p m e nt t hr es h ol d t e m p er at ur e, pl a nts d o n't gr o w.
B e y o n d T m a x/i ,  t h e  m a xi m u m  d e v el o p m e nt  t hr e s h ol d  t e m p er at ur e,  all  gr o wt h  c e as es
a g ai n 9 .  I us e d t h e f oll o wi n g e q u ati o ns t o d es cri b e t h e r el ati o ns hi p bet w e e n gr o wt h r at e r i
a n d t e m p er at ur e T (s. a. Fi g. 1):

ri  =   

⎩⎪
⎨
⎪⎧ rm a x

T – T mi n/i

T o pt/i – T mi n/i
T mi n/i < = T < = T o pt/i ∀ i s p e ci e s

rm a x

T o pt – T

T m a x /i – T o pt/i
T o pt/i < = T < = T m a x/i ∀ i s p e ci e s ( 2)

0 T < T mi n/i T m a x/i < T ∀ i s p e ci e s

 

w h er e

rm a x m a xi m u m gr o wt h r at e of d aisi es

T mi n/i mi ni m u m t hr es h ol d t e m p er at ur e of d e v el o p m e nt f or s p e ci es i

T m a x/i m a xi m u m t hr es h ol d t e m p er at ur e of d e v el o p m e nt f or s p e ci es i

T o pt/i o pti m u m gr o wt h t e m p er at ur e f or s p e ci es i

T h e m a xi m u m gr o wt h r at e a n y d ais y p o p ul ati o n c a n r e a c h i s gi v e n b y p ar a m et er r m a x .
T his is b e c a us e t h e t er m wit hi n p ar e nt h es es i n e q u ati o n ( 2) n e v er b e c o m es l ar g er t h a n 1
w hi c h  it  r e a c h e s  e x a ctl y  w h e n  T = T o pt/i  a n d  b e c o m e s  e x a ctl y  z er o  w h e n  T = T mi n/i
r es p e cti v el y T = Tm a x/i  ( Fi g. 1).  

                                                
9 Alt h o u g h  e xtr e m e  t e m p er at ur es  ar e  n ot  n e c ess aril y  l et h all y  c ol d  r es p e cti v el y  l et h all y  h ot,  t h e y  w o n't
all o w d aisi es t o e xist p er m a n e ntl y.  T h e p er m a n e nt s ur vi v al of a n y pl a nt s p e ci es r e q uir es at  l e ast  d uri n g a
mi ni m u m lif e- s p a n s o m e gr o wt h.   T h er ef or e, w e m a y i nt er pr et e a n y m e a n gl o b al t e m p er at ur e T  s m all er
t h a n Tmi n/i  r es p e cti v el y bi g g er t h a n Tm a x/i  t o dis all o w f or t h e e xist e n c e of d aisi es.  

T
mi n /i T m a x /i

T
o p t /i

r o

r

T [° C ]

i

Fi g. 1:  T e m p er at ur e d e p e n d e n c e of t h e p er c a pit a gr o wt h r at e r i of a p arti c ul ar d ai s y
s p e ci e s i.  r m a x  ;  T - gl o b al t e m p er at ur e o n  d ais y w orl d;   To pt/i ,  Tmi n/i ,  Tm a x/i  ar e  t h e
s p e ci e s s p e cifi c gr o wt h p ar a m et ers d efi ni n g wit hi n  w hi c h r a n g e d aisi es gr o w a n d w hi c h
t e m p er at ur e c orr es p o n ds t o o pti m al gr o wt h c o n diti o ns.

T h e li g ht er a d ais y, t h e b ett er h e at a d a pt e d  it is, h e n c e t h e  hig h er t h e p ar a m et ers T mi n/i ,
T o pt/i , a n d Tm a x/i  a n d vi c e v ers a .  Si n c e t h es e p ar a m et ers ar e s p e ci es s p e cifi c, it r es ults f or
e a c h s p e ci es a s p e cifi c, t e m p er at ur e T d e p e n d e nt gr o wt h r at e r i( T).  T hi s gi v es t h e hi g h
al b e d o d aisi es a c h a n c e t o c o ol a n d t h e l o w al b e d o d ar k o n es t o w ar m t h e d ais y w orl d.  

H o w e v er, t h e e q u ati o n ( 1) h as a si g nifi c a nt fl a w:  O n c e a p arti c ul ar s p e ci es c o m p os iti o n
h as b e e n est a blis h e d, pr o p orti o ns r e m ai n t h e s a m e f or e v er.  T his i s  b e c a us e n o  c h a n g es
t a k e pl a c e i n p o p ul ati o n si z e s o n c e t h e p o p ul ati o n t ot al of all s p e ci e s h a s  r e a c h e d t h e
c arr yi n g c a p a cit y;  i n t his c as e t h e e q u ati o n

K = x j∑
j = 1

n

( 3)

h ol ds a n d h e n c e

x i= 0          ∀ i ( 4)

a st at e w hi c h c orr es p o n ds t o a tr u e e q uili bri u m st at e.   I n  or d er  t o all o w f or c h a n g es i n
s p e ci e s c o m p o siti o n e v e n w h e n t h er e i s n o m or e fr e e s p a c e a v ail a bl e, I h a v e m o difi e d
e q u ati o n ( 1) i n t his w a y:

x i=r i( T)
K- x j∑

j = 1

n

K
x i - mix i ∀ i s p e ci es ( 5)

w h er e m i d e n ot es a m ort alit y r at e of t h e i-t h s p e ci es a n d all ot h er p ar a m et ers a n d v ari a bl es
h a v e t h e s a m e m e a ni n g a s i n ( 1).  T hi s c a n b e  i nt er pr et e d a s  a  c o nti n u o us, i nt ers p e cifi c
c o m p etiti o n a m o n g t h e d aisi es.  At e q uili bri u m n o w n ot e q u ati o n ( 3) b ut t h e f oll o wi n g
s p e ci es s p e cifi c c o n diti o n h ol ds

K 1- m i

ri
= x j∑

j = 1

n

( 3')

E q u ati o n ( 3') c a n b e i nt er pr et e d s u c h, t h at t h e c arr yi n g c a p a cit y is r e d u c e d f or e a c h s pe ci es
b y t h e pr o p orti o n p i= m i/ri.   If t h er e i s at l e a st o n e pi, w hi c h h a s a v al u e s m all er a n d
diff er e nt fr o m all ot h er p i's, t his p arti c ul ar s p e ci es is still all o w e d t o gr o w, w hil e all ot h er
h a v e alr e a d y r e a c h e d z er o gr o wt h c o n diti o n, i. e. e x c e pt f or s p e ci e s  i, c o n diti o n ( 3')  i s
s atisfi e d.  A c c or di n g t o e q u ati o n ( 1) or ( 5) gr o wt h r at es b e c o m e n e g ati v e if t h e ri g ht- h a n d
p arts of e q u ati o n ( 3) r es p e cti v el y ( 3') b e c o m e l ar g er t h a n t h e l eft- h a n d p arts.  T his m e a ns

t h at a still gr o wi n g s p e ci es wit h a l o w pi c a n  t a k e  a w a y  s p a c e  fr o m  a n ot h er  s p e ci es  b y
r e d u ci n g its p o p ul ati o n, gi v e n t h at t his s p e ci es h as alr e a d y c e as e d t o gr o w b e c a us e of its
bi g g er p i.   T h e  i m p ort a nt  c h ar a ct eri sti c  of  t h e  w h ol e  m o d el  ( 5)  r e s ulti n g  fr o m  t his
b e h a vi o ur is t h at e v e n o n c e e q uili bri u m h as b e e n r e a c h e d, c h a n g es i n t h e e n vir o n m e nt c a n
alt er t h e e q uili bri u m c o n diti o n a n d t h us s hift t h e e q uili bri u m st at e a w a y fr o m  its  c urr e nt
p ositi o n.  T h e l att er w o ul d n ot b e p ossi bl e f or m o d el ( 1).

C o n c er ni n g t h e d ai s y b e h a vi o ur t h e m o d el i s n o w c o m pl et e.   W h at  still r e m ai n s i s  a
r el ati o ns hi p b et w e e n t h e gl o b al al b e d o ag  as gi v e n b y

a g = 1
K

a ix i∑
i = 1

n

 + ab ( K- xi∑
i = 1

n

)
( 6)

w h er e
a i  al b e d o of i-t h s p e ci es

a b   al b e d o of b ar e gr o u n d.

a n d t h e m e a n gl o b al t e m p er at ur e T.  T his r el ati o ns hi p m ust b e n e g ati v el y c orr el at e d, a n d
a g ai n f oll o wi n g p arsi m o n y, I us e d a li n e ar r el ati o ns hi p of t h e f or m:

T = ( T #
m a x – T #

mi n ) ( 1 – ag ) (7 )

w h er e a g  is gi v e n b y e q u ati o n ( 6) a n d T is t h e ass o ci at e d gl o b al t e m p er at ur e ( Fi g. 2).  T h e
p ar a m et er s T *

mi n  a n d  T *
m a x  ar e  t h e  t e m p er at ur e s  at  w hi c h  t h e  al b e d o  i s  m a xi m al

r es p e cti v el y mi ni m al.

mi n

*

T

m a x

*

T

T [° C ]

1 - a g0 1

Fi g. 2:  Li n e ari z e d r el ati o ns hi p b et w e e n t h e m e a ns of gl o b al t e m p er at ur e a n d t h e gl o b al
al b e d o a g  o n d ais y w orl d.  At a n y m o m e nt ag  is c al c ul at e d fr o m e q u ati o n ( 6) w hi c h  d e p e n ds
o n t h e c urr e nt si z e s of d ai s y p o p ul ati o n s a n d t h e pr o p orti o n of u n p o p ul at e d, b ar e gr o u n d.
T h e al b e d o t h e n d et er mi n e s t h e gl o b al  t e m p er at ur e T  v ar yi n g  wit hi n  r a n g e T *

mi n  a n d
T *

m a x .

A c c or di n g t o t h e b asi c ass u m pti o n t h e m o d el s h o ul d als o all o w t o  gr a d u all y c h a n g e t h e
t e m p er at ur e r e gi m e b y i n cr e a si n g t h e s ol ar l u mi n o sit y t hr o u g h it s a g ei n g pr o c e s s.  T o
o bt ai n t his b e h a vi o ur I  h a v e i ntr o d u c e d a n  a d diti o n al st at e v ari a bl e s, a  stri ctl y p os iti v e
a n o m al y  w hi c h  a c c o u nt s  f or  t h e  t e m p er at ur e  i n cr e a s e  d u e  t o  t h e  c h a n gi n g  s ol ar
l u mi n o sit y.   I  a s s u m e d  t h at  t h e  t e m p or al  b e h a vi o ur  of  t hi s  eff e ct  i s  gi v e n  b y  t his
diff er e nti al e q u ati o n

s  = σ ( 8)

w h er e σ  is a c o nst a nt.  E q u ati o n ( 7) t h e n als o t o b e m o difi e d t o

T = ( T *
m a x – T *

mi n ) ( 1 – ag ) + ( s- so ) (7 ')

T h e  m o d el  i s  n o w c o m pl et e, a n d t h e e q u ati o n s ( 2), ( 5), ( 6), ( 7'), a n d ( 8)  d e s cri b e all
d y n a mi cs o n d ais y w orl d.  I n a d d iti o n t h e p ara m et er e q u ati o ns ( 2 a), ( 2 b) m ust b e s atisfi e d
a n d a n o n-tri vi al e q uili bri u m st at e f or a p arti c ul ar s ol ar l u mi n o sit y i s d et er mi n e d b y
e q u ati o n ( 3').  T h e m o d el d es cri b es i n p arti c ul ar t h e c h a n g es i n d ais y p o p ul ati o ns gi v e n b y
t h e drifti n g e q uili bri u m st at e as it is d et er mi n e d b y t h e a g ei n g of t h e su n.  It a d h er es i n all
as p e cts t o t h e b asi c ass u m pti o ns of t h e G ai a h y p ot h esis.

P a r a m et e r e sti m at es

O nl y a f e w m o d el p ar a m et ers h a v e b e e n gi v e n b y L O V E L O C K  ( 1 9 8 8), t h e m aj orit y h a d t o
b e esti m at e d fr o m k n o w n p h ysi c al pr o p erti es, s u c h as c urr e nt t e m p er at ur e r e gi m es et c.  

Si n c e t h e m o d el e q u ati o ns ( 2), ( 5), ( 6),  ( 7'), a n d ( 8) f or m a c o ntin u o us ti m e s y st e m of
or di n ar y, n o n-li n e ar diff er e nti al e q u ati o ns, t h e b asi c ti m e u nit m ust b e d efi n e d b ef or e a n y
ot h er p ar a m et er s c a n b e d et er mi n e d.  T h e s ol ar  l u mi n o sit y  i n cr e a s e s  fr o m  0. 6  t o  1. 4
r el ati v e t o w h at it is t o d a y i n a b o ut 8⋅1 0 9  y e ar s.   A c c or di n gl y, t h e p eri o d of i nt er est f or
w hi c h t h e e q u ati o ns o u g ht t o b e s ol v e d li es i n- b et w e e n t h e ti m e p oi nts t o  = 0 a n d te n d  =
8 ⋅1 0 9  y e ars.

P ar a m et er s y m b ol u nit v al u e

R at e of i n cr e as e of t e m p er at ur e d u e t o t h e i n cr e as e
of s ol ar l u mi n osit y

σ ° C 4 ⋅1 0 - 9 6

I niti al s ol ar l u mi n osit y i n % of s ol ar c o nst a nt s o % 6 0. 0

Gl o b al c arr yi n g c a p a cit y f or d aisi es K # 1. 0

M a xi m u m p er c a pit a d ais y gr o wt h r at e rm a x 4 ⋅1 0 - 9 6 0. 0

Gl o b al t e m p er at ur e at al b e d o = 1 T *
mi n ° C 0. 0

Gl o b al t e m p er at ur e at al b e d o = 0 T *
m a x ° C 3 0. 0

M ort alit y of li g ht d aisi es m[li g ht] 4 ⋅1 0 - 9 4 0. 0

M ort alit y of d ar k d aisi es m[ d ar k] 4 ⋅1 0 - 9 8 0. 0

Mi ni m u m t e m p er at ur e f or gr o wt h of li g ht d aisi es  T mi n [li g ht] ° C 1 5. 0

Mi ni m u m t e m p er at ur e f or gr o wt h of d ar k d aisi es  T mi n [ d ar k] ° C 5. 0

T e m p er at ur e f or o pti m u m gr o wt h of li g ht d aisi es  T o pt [li g ht] ° C 3 2. 5

T e m p er at ur e f or o pti m u m gr o wt h of d ar k d aisi es  T o pt [ d ar k] ° C 2 2. 5

Al b e d o of li g ht d aisi es a[li g ht] - 0. 7

Al b e d o of d ar k d aisi es a[ d ar k] - 0. 2

Al b e d o of b ar e gr o u n d a b - 0. 4

T a b. 1: P ar a m et ers us e d t o n u m eri c all y s ol v e t h e d ais y w orl d m o d el gi v e n b y  t h e  e q u ati o ns  ( 2),  ( 5),  ( 6),
( 7'), a n d ( 8).

T h e i niti al st at e v e ct or us e d w as: x o ' = [ 0. 0 1, 0. 0 1] a n d so  = 6 0.  T h e i nt e gr ati o n m et h o d
w as a v ari a bl e st e p l e n gt h R u n g e- K utt a 5t h or d er al g orit h m.

T h e f oll o wi n g p ar a m et ers h a v e b e e n gi v e n b y L O V E L O C K ( 1 9 8 8) f or a t w o s p e ci es m o d el:
li g ht d aisi es h a v e a n al b e d o of 0. 7, d ar k d aisi es o n e of 0. 2.  B ar e gr o u n d is d es cri b e d t o
h a v e a n al b e d o of 0. 4.  

T h e r e m ai ni n g p ar a m et ers h a v e b e e n d eri v e d t h e or eti c all y or sli g htl y a dj ust e d t o o p er at e i n
t h e s a m e r a n g e of s y st e m b e h a vi o ur a s d e s cri b e d b y LO V E L O C K  ( 1 9 8 8).    N o  tr u e
p ar a m et er i d e ntifi c ati o n pr o c e d ur e h as b e e n a p pli e d.  

Si m ul ati o n r e s ult s

T o s ol v e t h e m n u m eri c all y, t h e s yst e m of diff er e nti al e q u ati o ns d efi n e d b y E q. ( 2), ( 5), ( 6),
( 7'),  a n d  ( 8)  h a v e  b e e n  i m pl e m e nt e d  a s  a  M o d el W or k s  m o d el  d efi niti o n  pr o gr a m
( FI S C H LI N et al ., 1 9 9 3; s e e A p p e n di x).  U si n g t h e p ar a m et er v al u e s gi v e n i n T a b. 1  I
o bt ai n e d f or j ust t w o d ais y s p e ci es, a li g ht a n d a d ar k o n e, t h e r es ults s h o w n i n Fi g. 3.  

s l
 7 0 . 0  8 0 . 0  9 0 . 0  1 0 0 . 0  1 1 0 . 0  1 2 0 . 0  1 3 0 . 0  1 4 0 . 0

 0 . 0

 0 . 2

 0 . 4

 0 . 6

 0 . 8

 1 . 0

C u r v e s M i n i m u m M a x i m u m U n i t
x [ 1 ]        0 . 0 0 0       1 . 0 0 0 #
x [ 2 ]        0 . 0 0 0       1 . 0 0 0 #
T       0 . 0 0 0      7 0 . 0 0 0 ° C
T '       0 . 0 0 0      7 0 . 0 0 0 ° C

S t a n d a r d  r u n  D a i s y w o r l d  -  M o d e l  b y  A n d r e a s  F i s c h l i n

C u r v e s M i n i m u m M a x i m u m U n i t
x [ 1 ]        0 . 0 0 0       1 . 0 0 0 #
x [ 2 ]        0 . 0 0 0       1 . 0 0 0 #
T       0 . 0 0 0      7 0 . 0 0 0 ° C
T '       0 . 0 0 0      7 0 . 0 0 0 ° C

S t a n d a r d  r u n  D a i s y w o r l d  -  M o d e l  b y  A n d r e a s  F i s c h l i n

Fi g. 3: Si m ul ati o n r e s ult s of t h e d eri v e d d ai s y w orl d m o d el ( E q. ( 2), ( 5), ( 6), ( 7'), a n d
( 8)).  It r e s e m bl e s q u alit ati v el y t h e r e s ult s d e pi ct e d i n Fi g. 2. 1 i n L o v el o c k' s b o o k  « T h e
A g e s of G ai a » ( L O V E L O C K , 1 9 8 8), i. e. t w o d ai s y s p e ci e s, a d ar k a n d a li g ht o n e, ar e
p o p ul ati n g  d ai s y  w orl d  a n d  ar e  c a p a bl e  of  r e g ul ati n g  h o m o e o st ati c all y  t h e   gl o b e's
t e m p er at ur e T.  T h e gr a p h s h o w s al s o w h at t h e m e a n gl o b al  t e m p er at ur e T'  w o ul d h a v e
b e e n  wit h o ut  a n y  d ai si e s  o n  a  lif e-l e s s  pl a n et.   L e g e n d:   x[ 1],  x[ 2]  -  si z e s  of  li g ht
r e s p e cti v el y  d ar k  d ai s y  p o p ul ati o n s;  T,  T'  -  m e a n  gl o b al  t e m p er at ur e s  i n  pr e s e n c e
r e s p e cti v el y  a b s e n c e  of  d ai si e s;   -  r a n g e  of  t e m p er at ur e  [ T o pt/ 1 .. To pt/ 2 ]
e n c o m p a s si n g o pti m al gr o wt h c o n diti o n s f or all  d ai s y s p e ci e s; sl  -  s ol ar l u mi n o sit y  i n
p er c e nts of t o d a y's s ol ar l u mi n osit y.  

I n t his m o d el it is e as y t o h a v e a lif e-l ess pl a n et b y si m pl y s etti n g t h e i niti al st at e ve ct or x o '
= [ 0. 0, 0. 0], t h e tri vi al st e a d y st at e of all lif e ( Fi g. 3, c ur v e T').   If o n e c o m p ar e s t h e
t e m p er at ur e T wit h T', i. e. t h e t e m p er at ur e o n t h e gl o b e wit h r es p e cti v el y wit h o ut d aisi es,
t h e h o m e ost asis pr o p ert y cl e arl y e m er g es, si n c e t h e d aisi es o n  t h e  pl a n et ar e  c a p a bl e t o
r e g ul at e t e m p er at ur e cl os e t o t h eir o pti m u m gr o wt h c o n diti o ns ( Fi g. 3, c ur v e T).  O n a lif e-
l ess pl a n et t h at r e g ul ati o n w o ul d b e missi n g a n d w o ul d s h o w j ust a r a m p, i. e. a c c or di n g t o
E q. ( 8) t h e c o nst a nt i n cr e as e of t e m p er at ur e T' as d ais y w orl d's s u n b e c o m es ol d er.  

S t a n d a r d  r u n  D a i s y w o r l d  m o d e l  b y  A n d r e a s  F i s c h l i n

Fi g. 4:  E m piri c al i n v e sti g ati o n of t h e st a bilit y of t h e d ai s y w orl d m o d el ( E q. ( 2),  ( 5),
( 6), ( 7'), a n d ( 8)) u n d er r a n d o ml y i m p a cti n g  m et er orit e s.   D e p e n di n g o n  t h e  si z e  of  t h e
m et e orit e, a v ar yi n g fr a cti o n of t h e d aisi es ar e d estr o y e d a n d t h e  t e m p er at ur e  o n  d ais y  w orl d
i s  di st ur b e d.    U nl e s s  all  d ai si e s  ar e  c om pl et el y  wi p e d  o ut,   t h e  s y st e m s  c a p a bilit y  t o
r e g ul at e t h e  gl o b e' s  t e m p er at ur e  T  re m ai ns.  T h e gr a p h s h o ws als o w h at t h e m e a n gl o b al
t e m p er at ur e T' w o ul d h a v e b e e n wit h o ut a n y d aisi es o n a lif e-l ess pl a n et.    L e g e n d:    T,  T'  -
m e a n gl o b al t e m p er at ur es i n pr es e n c e r es p e cti v el y a bs e n c e of d aisi es.  

First t h e b e h a vi o ur of t h e m o d el is q u alit ati v el y s i mil ar t o t h at s h o w n i n L o v el o c k's b o o k
( LO V E L O C K , 1 9 8 8; e. g. Fi g. 2. 1).  T h e d aisi es ar e c a p a bl e first t o i n cr e as e t h e t e m p er at ur e
T a b o v e w h at it w o ul d  b e  wit h o ut  t h e  d aisi es.  T his c o n diti o n is cl os er t o t h e t e m p er at ur e
f or o pti m u m gr o wt h of t h e d aisi es ( T > T', s e e l eft p art of Fi g. 3).  T h e pl a n et is d o mi n at e d
b y d ar k d aisi es, w hi c h will gi v e u p t his p ositi o n o nl y if t h e s ol ar l u mi n osit y h as i n cr e as e d
s o m u c h, t h at t h e t e m p er at ur e b e c o m es m u c h t o o h ot f or t h e d ar k d aisi es, gi vi n g pl a c e t o
t h e li g ht d aisi es' c o oli n g c a p a bilit y.   T h e l att er s uc c e e d i n c o oli n g t h e t e m p er at ur e n o w
b el o w w h at it w o ul d h a v e b e e n o n a lif e-l ess pl a n et, t h us p ost p o ni n g t h e c oll a ps e of all lif e
( T < T', s e e mi d dl e p art of Fi g. 3).  If t h e s ol ar l u mi n osit y h as r e a c h e d s u c h gr e at v al u es as
t o c a u si n g t h e t e m p er at ur e t o b e t o o h ot e v e n f or  t h e li g ht d aisi es, d ai s y p o p ul ati o ns
dis a p p e ar a n d a n y t e m p er at ur e r e g ul ati o n b y lif e c oll a ps es, i n di c at e d b y t h e m er gi n g of t h e
t w o c ur v es T a n d T'  ( T≈ T', s e e ri g ht p art of Fi g. 3).

Dis c ussi o n a n d c o n cl usi o ns

S e c o n d t his i n v esti g ati o n s h o ws t h at t h e m ai n p oi nt of L o v el o c k's G ai a h y p ot h esis, i. e. t h e
h o m e ost ati c c a p a bilit y of lif e t o r e g ul at e gl o b a l t e m p er at ur e wit hi n a r a n g e f a v o ur a bl e f or
li vi n g or g a ni s m s, c a n b e r e pr o d u c e d b y u si n g m o d el s w hi c h ar e diff er e nt a n d  d eri v e d
i n d e p e n d e ntl y  fr o m  t h at  f or m ul at e d  b y  WA T S O N  a n d  LO V E L O C K  ( 1 9 8 3).    H e n c e
L o v el o c k's c o n cl usi o n a p p e ars n ot t o d e p e n d s ol el y o n t h e m o r e or l ess ar bitr ar y c h os e n
e q u ati o n s w hi c h m o d el t h e c o m p etiti o n of t h e d ai s y p o p ul ati o n s ( C A R T E R & P RI N C E ,
1 9 8 1), b ut  t o b e r o b ust e n o u g h f or a wi d e n u m b er of c o m p etiti o n m o d els.  H o w e v er, t h e
l att er o u g ht still t o b e e x pl or e d f urt h er.

I failed to falsify the most interesting part of the Gaia hypothesis, i.e. life's capability of
homeostasis, nevertheless, its other parts can either be shown to be almost trivial, i.e. the
analogy between an ecosystem and the ecosphere, or to be merely the result of a subjective
redefinition of the term life.  At least to me it appears, that his hypothesis would deserve
more attention by serious scientists, than it is likely to receive because of Lovelock's way
to reason without persistent rigor throughout all his arguments.  Maybe that's the price to
pay for thought provoking ideas...

Appendix

The here described model has been implemented, numerically solved, and its behavior
studied by the following simulation model.  It represents a ModelWorks model definition
program (FISCHLIN et al., 1993) written in Modula-2 (WIRTH, 1985).

MODULE Daisyworld;

  (*******************************************************************

    ModelWorks model:  Daisyworld

          Copyright ©1989 by Andreas Fischlin and Swiss
          Federal Institute of Technology Zurich ETHZ
          Department of Environmental Sciences
          Systems Ecology Group
          ETH-Zentrum
          CH-8092 Zurich
          Switzerland

      Purpose Models the daisyworld, which demondescrStrates, according
        to its author Lovelock the essence of the
        Gaia-hypothesis: The temperature regime on a globe
        covered by a single continent is regulated by daisy
        species differing in their color from dark to light and
        thus influencing the albedo of the globe.  

      References

        Lovelock, J. 1988. The ages of gaia. New York and London:
        W.W. Norton & Co., Inc. 252pp. ISBN 0-393-02583-7.

        Implementation and Revisions:
        ============================

        Author  Date        Description
        ------  ----        -----------

        af      07/03/90    First implementation (MacMETH 2.6+,
        DM 2.0, MW 1.3a)

  *******************************************************************)

  FROM DMStrings IMPORT
    Concatenate;
  FROM DMConversions IMPORT
    IntToString;
  FROM DMMenus IMPORT
    Menu, Command, AccessStatus, Marking, InstallMenu,
    InstallCommand, InstallAliasChar;  

  FROM SimBase IMPORT
    Model, DeclM, NoOutput, IntegrationMethod, DeclSV, RTCType,
    DeclP, NoTerminate, StashFiling, DeclMV, Tabulation, NoAbout,
    Graphing, SetDefltCurveAttrForMV, Stain, LineStyle,
    SetIntegrationStep, SetMonInterval, SetSimTime,
    SetDefltProjDescrs;  
  FROM SimMaster IMPORT
    RunSimEnvironment, CurrentTime;   

  FROM RandGen IMPORT U;

 

  TYPE
    Color = (light,dark);
   

  VAR
    m: Model;
    daisy: ARRAY [MIN(Color)..MAX(Color)] OF
    RECORD
     x,xDot,
     albedo,mu,optTemp,minTemp, r,m
     : REAL;
    END(*RECORD*);
    r0,K,

    albedoGlob,sumOfx,albedoB,
    Temp,TempDash, TempMaxStar,TempMinStar,
    solLum,solLumDeltaT,solLumDeltaTDot,sigma,
    meteorDeltaT,
    clipGrowthR: REAL;
    clipGrowth: BOOLEAN;

  PROCEDURE Input;
    VAR c: Color;
  BEGIN
    sumOfx := 0.0;
    FOR c:= MIN(Color) TO MAX(Color) DO
      sumOfx := sumOfx+daisy[c].x;
    END(*FOR*);
    albedoGlob := 0.0;
    FOR c:= MIN(Color) TO MAX(Color) DO
      albedoGlob := albedoGlob + daisy[c].albedo*daisy[c].x/K;
    END(*FOR*);
    albedoGlob := albedoGlob + (K-sumOfx)/K*albedoB;
    Temp := TempMaxStar-(TempMaxStar-TempMinStar)*albedoGlob+solLumDeltaT;
    TempDash := TempMaxStar-(TempMaxStar-TempMinStar)*albedoB+solLumDeltaT;
    FOR c:= MIN(Color) TO MAX(Color) DO
      WITH daisy[c] DO
        r := r0*(mu*(Temp-optTemp)*(Temp-optTemp)+1.0);
        IF (r<=0.0) AND clipGrowth THEN r := 0.0 END;
      END(*WITH*);
    END(*FOR*);
    solLum := CurrentTime();
  END Input;
 

  PROCEDURE Initialize;
    VAR c: Color;
  BEGIN
    FOR c:= MIN(Color) TO MAX(Color) DO
      WITH daisy[c] DO
        mu := -1.0/((minTemp-optTemp)*(minTemp-optTemp));
      END(*WITH*);
    END(*FOR*);
    clipGrowth := clipGrowthR<>0.0;
  END Initialize;

  PROCEDURE Dynamic;
    VAR c: Color;
  BEGIN
    FOR c:= MIN(Color) TO MAX(Color) DO
      WITH daisy[c] DO
        xDot := r*((K-sumOfx)/K)*x - m*x;
      END(*WITH*);
    END(*FOR*);
    solLumDeltaTDot := sigma;
  END Dynamic;

  PROCEDURE DeclModelObjects;
    VAR descrStr,symStr,istr: ARRAY [0..127] OF CHAR; c: Color;
      nrSpecies,nrIntervals: REAL;
  BEGIN
    nrSpecies := FLOAT(ORD(MAX(Color))+1); nrIntervals := nrSpecies-1.0;
    DeclSV(solLumDeltaT,solLumDeltaTDot,0.0,MIN(REAL),MAX(REAL),"Solar luminosity","s","%");
    DeclP(sigma,0.6,MIN(REAL),MAX(REAL),rtc,
          "Rate of increase of solar luminosity", "sigma","/time");
    DeclP(K, 1.0, 0.0, MAX(REAL), rtc,
      'Global carrying capacity of globe', 'K', '#');
    DeclP(r0, 3.0, MIN(REAL), MAX(REAL), rtc,
      "Daisy growth rate", "r0", '/time');
    DeclP(clipGrowthR, 1.0, 0.0, 1.0, rtc,
      "Flag to clip growth outside Tmin and Tmax", "clipGrowth", '0 or 1');
    DeclP(albedoB, 0.4, 0.0,1.0, rtc,
      "Albedo of bare, unpopulated ground", "ab", '');
    DeclP(TempMinStar, 0.0, MIN(REAL), MAX(REAL), rtc,
      "Temperature at albedo minimum", "T*min", '°C');
    DeclP(TempMaxStar, 30.0, MIN(REAL), MAX(REAL), rtc,
      "Temperature at albedo maximum", "T*max", '°C');
    DeclP(meteorDeltaT, 20.0, MIN(REAL), MAX(REAL), rtc,
      "Meteorite impact temperature jump", "∆Tm", '°C');

    DeclMV(albedoGlob,0.0,1.0,'Global albedo','ag','',
      notOnFile,writeInTable,notInGraph);
    DeclMV(daisy[light].r,-1.0,1.0,'Growth rate','r[1]','/time',
      notOnFile,notInTable,notInGraph);
    DeclMV(daisy[dark].r,-1.0,1.0,'Growth rate','r[2]','/time',
      notOnFile,notInTable,notInGraph);
    DeclMV(sumOfx,0.0,K,'Global daisy population','∑x','#',
      notOnFile,notInTable,notInGraph);
    DeclMV(Temp,0.0,70.0,'Global temperature','T','°C',
      notOnFile,writeInTable,isY);
    SetDefltCurveAttrForMV(m,Temp,sapphire,unbroken,0C);
    DeclMV(TempDash,0.0,70.0,'Global temperature without life',"T'",'°C',
      notOnFile,writeInTable,isY);
    SetDefltCurveAttrForMV(m,TempDash,turquoise,spotted,0C);
    DeclMV(solLum,60.0,140.0,'Solar luminosity','sl','%',
      notOnFile,writeInTable,isX);

    FOR c:= MIN(Color) TO MAX(Color) DO
      WITH daisy[c] DO
        descrStr := 'Population of ';
        IF c=MIN(Color) THEN
          Concatenate(descrStr,"light ",descrStr);
        ELSIF c=MAX(Color) THEN
          Concatenate(descrStr,"dark ",descrStr);
        ELSE
          Concatenate(descrStr,"intermediate ",descrStr);
        END(*IF*);
        Concatenate(descrStr,"daisies",descrStr);
        symStr := "x[";
        IntToString(1+ORD(c),istr,0);
        Concatenate(symStr,istr,symStr);
        Concatenate(symStr,"] ",symStr);
        DeclSV(x, xDot,0.01, 0.0, K, descrStr, symStr, '#');
        DeclMV(x,0.0,K, descrStr,symStr,'#',
               notOnFile,writeInTable,isY);
      END(*WITH*);

      IF c=MIN(Color) THEN
        SetDefltCurveAttrForMV(m,daisy[c].x,gold,dashSpotted,0C);
      ELSIF c=MAX(Color) THEN
        SetDefltCurveAttrForMV(m,daisy[c].x,coal,broken,0C);
      ELSE
        SetDefltCurveAttrForMV(m,daisy[c].x,emerald,autoDefStyle,0C);
      END(*IF*);
    END(*FOR*);
       

    FOR c:= MIN(Color) TO MAX(Color) DO
      WITH daisy[c] DO
        descrStr := "Mortality of ";
        IF c=MIN(Color) THEN
          Concatenate(descrStr,"light ",descrStr);
        ELSIF c=MAX(Color) THEN
          Concatenate(descrStr,"dark ",descrStr);
        ELSE
          Concatenate(descrStr,"intermediate ",descrStr);
        END(*IF*);
        Concatenate(descrStr,"daisies",descrStr);
        symStr := "m[";
        IntToString(1+ORD(c),istr,0);
        Concatenate(symStr,istr,symStr);
        Concatenate(symStr,"] ",symStr);
        DeclP(m, 0.2+0.2/nrIntervals*FLOAT(ORD(c)), MIN(REAL),MAX(REAL), rtc,
          descrStr, symStr, '/time');
      END(*WITH*);
    END(*FOR*);
       

    FOR c:= MIN(Color) TO MAX(Color) DO
      WITH daisy[c] DO
        descrStr := "Temperature minimum for growth of ";
        IF c=MIN(Color) THEN
          Concatenate(descrStr,"light ",descrStr);
        ELSIF c=MAX(Color) THEN
          Concatenate(descrStr,"dark ",descrStr);
        ELSE
          Concatenate(descrStr,"intermediate ",descrStr);
        END(*IF*);
        Concatenate(descrStr,"daisies",descrStr);
        symStr := "Tmin[";
        IntToString(1+ORD(c),istr,0);
        Concatenate(symStr,istr,symStr);
        Concatenate(symStr,"] ",symStr);
        DeclP(minTemp, 15.0-10.0/nrIntervals*FLOAT(ORD(c)), MIN(REAL), MAX(REAL), rtc,
          descrStr, symStr, '°C');
      END(*WITH*);
    END(*FOR*);
       

    FOR c:= MIN(Color) TO MAX(Color) DO
      WITH daisy[c] DO
        descrStr := "Temperature optimum for growth of ";
        IF c=MIN(Color) THEN
          Concatenate(descrStr,"light ",descrStr);
        ELSIF c=MAX(Color) THEN
          Concatenate(descrStr,"dark ",descrStr);
        ELSE
          Concatenate(descrStr,"intermediate ",descrStr);
        END(*IF*);
        Concatenate(descrStr,"daisies",descrStr);
        symStr := "Topt[";
        IntToString(1+ORD(c),istr,0);
        Concatenate(symStr,istr,symStr);
        Concatenate(symStr,"] ",symStr);
        DeclP(optTemp, 32.5-10.0/nrIntervals*FLOAT(ORD(c)), MIN(REAL), MAX(REAL), rtc,
          descrStr, symStr, '°C');
      END(*WITH*);
    END(*FOR*);
       

    FOR c:= MIN(Color) TO MAX(Color) DO
      WITH daisy[c] DO
        descrStr := 'Albedo of ';
        IF c=MIN(Color) THEN
          Concatenate(descrStr,"light ",descrStr);
        ELSIF c=MAX(Color) THEN
          Concatenate(descrStr,"dark ",descrStr);
        ELSE
          Concatenate(descrStr,"intermediate ",descrStr);
        END(*IF*);
        Concatenate(descrStr,"daisies",descrStr);
        symStr := "a[";
        IntToString(1+ORD(c),istr,0);
        Concatenate(symStr,istr,symStr);
        Concatenate(symStr,"] ",symStr);
        (* albedos range from 0.2 (dark daisy) to 0.7 (light daisy) *)
        DeclP(albedo, 0.7-0.5/nrIntervals*FLOAT(ORD(c)),
              0.0, 1.0, rtc, descrStr, symStr, '');
      END(*WITH*);
    END(*FOR*);      

  END DeclModelObjects;

  VAR
    myMenu: Menu; impactCmd: Command;
   

  PROCEDURE MeteorImpact;
    VAR c: Color;
    PROCEDURE Sign(x: REAL): REAL;
    BEGIN
      IF x<0.0 THEN RETURN -1.0 ELSE RETURN 1.0 END
    END Sign;
  BEGIN
    Temp := Temp - Sign(U()-0.5)*meteorDeltaT + (U()-0.5);
    FOR c:= MIN(Color) TO MAX(Color) DO
      WITH daisy[c] DO
        x := U()*x
      END(*WITH*);
    END(*FOR*);
  END MeteorImpact;
   

  PROCEDURE AddCommands;
  BEGIN
    InstallMenu(myMenu,'Events',enabled);
    InstallCommand(myMenu, impactCmd,"Meteorite impact", MeteorImpact,
                   enabled, unchecked);

    InstallAliasChar(myMenu,impactCmd,"Z");
  END AddCommands;

  PROCEDURE DeclModel;
  BEGIN
    DeclM(m, RungeKutta4, Initialize, Input, NoOutput, Dynamic,
          NoTerminate, DeclModelObjects,
          'Daisyworld model and the Gaia hypothesis','m', NoAbout);
    SetSimTime(60.0,140.0);
    SetMonInterval(1.0); SetIntegrationStep(0.5);
    SetDefltProjDescrs(
      "Standard run Daisyworld - Model version by Andreas Fischlin","","",
      TRUE,FALSE,TRUE, TRUE,TRUE,TRUE,TRUE,TRUE);
    AddCommands;
  END DeclModel;

BEGIN
  RunSimEnvironment(DeclModel);
END Daisyworld .

